prog_start -> functions

functions -> function functions
functions -> epsilon

function -> FUNCTION ident SEMICOLON BEGIN_PARAMS declarations END_PARAMS
BEGIN_LOCALS declarations END_LOCALS BEGIN_BODY statements END_BODY

declarations -> declaration SEMICOLON declarations
declarations -> epsilon

statements -> statement SEMICOLON statements
statements -> epsilon

declaration -> identifiers COLON INTEGER
declaration -> identifiers COLON ARRAY L_SQUARE_BRACKET NUMBER
R_SQUARE_BRACKET OF INTEGER

identifiers -> ident COMMA identifiers
identifiers -> ident

statement -> var ASSIGN expression
statement -> IF bool_exp THEN statements else ENDIF
statement -> WHILE bool_exp BEGINLOOP statements ENDLOOP
statement -> DO BEGINLOOP statements ENDLOOP WHILE bool_exp
statement -> WRITE vars
statement -> READ vars
statement -> BREAK
statement -> RETURN expression

else -> ELSE statements
else -> epsilon

bool_exp -> relation_and_exp bool_exp_opt
bool_exp_opt -> OR relation_and_exp bool_exp_opt
bool_exp_opt -> epsilon

relation_and_exp -> relation_exp relation_and_exp_opt
relation_and_exp_opt -> AND relation_exp relation_and_exp_opt
relation_and_exp_opt -> epsilon

relation_exp -> NOT relation_exp
relation_exp -> expression comp expression
relation_exp -> TRUE
relation_exp -> FALSE
relation_exp -> L_PAREN bool_exp R_PAREN

expression -> multiplicative_expression expression_opt
expression_opt -> ADD multiplicative_expression expression_opt
expression_opt -> SUB multiplicative_expression expression_opt
expression_opt -> epsilon

multiplicative_expression -> term multiplicative_expression_opt
multiplicative_expression_opt -> MULT term multiplicative_expression_opt
multiplicative_expression_opt -> DIV term multiplicative_expression_opt
multiplicative_expression_opt -> MOD term multiplicative_expression_opt
multiplicative_expression_opt -> epsilon

comp -> EQ
comp -> NEQ
comp -> LT
comp -> GT
comp -> LTE
comp -> GTE

term -> SUB term2
term -> term2
term2 -> var
term2 -> NUMBER
term2 -> L_PAREN expression R_PAREN
term -> ident L_PAREN expressions R_PAREN

expressions -> expression COMMA expressions
expressions -> expression

vars -> var COMMA vars
vars -> var

var -> ident
var -> ident L_SQUARE_BRACKET expression R_SQUARE_BRACKET

identifiers -> ident COMMA identifiers
identifiers -> ident

ident -> IDENT ID